I Zählparameter in Schleifen
Das Beispielprogramm aus Bild 1 könnte in FORTH wie folgt programmiert werden:
Programmieren wir zuerst die Teilsumme S := S + TAB [I]:

```
S @ TAB  I + @ + S !
```

Nachdem dieses Teilprogramm im Dialog ausgetestet ist, wird ein neues FORTH-Wort definiert:

```
: TEILSUMME  S @
  TAB  I + @ + S ! ;
```

Nun wird die Schleife programmiert. DO verlangt im Stack den Endwert + 1 und den Anfangswert:

```
20 0 DO TEILSUMME LOOP
```

Nach der Testung wird definiert:

```
: SUMME  20 0 DO
  TEILSUMME  LOOP ;
```

Ähnlich verhält es sich mit der »bedingten Anweisung«: Vor dem Wort IF muß der Ausdruck für die Bedingung berechnet werden. Ist der Wert ungleich 0 (true), so werden die Worte zwischen IF und THEN ausgeführt. THEN schließt die bedingte Anweisung ab.

e) Ein- und Ausgabeoperationen
Es stehen eine Reihe von Worten für die verschiedenen Formen der Terminalein- und -ausgabe zur Verfügung. Eine Fileverarbeitung ist im FIG-FORTH nicht definiert. Deshalb sind in den verschiedenen FORTH-Systemen spezielle Wörterbücher für erweiterte E/A-Operationen vereinbart, die bei Bedarf mit dem Wort LOAD von einem Screen geladen werden.

Zur Erhöhung der Lesbarkeit der FORTH-Programme sollte man Kommentare ergänzen. Kommentare werden durch (eingeleitet und) abgeschlossen.
Diese kurzen Erläuterungen sollen hier als kleine Einführung in FORTH genügen. Die ausführliche Sprachbeschreibung ist der entsprechenden Fachliteratur zu entnehmen. (Informationen darüber können beim Autor eingeholt werden.)
Bild 8 zeigt das FORTH-Programm für die Textverarbeitungsaufgabe aus Heft 3. Das Wörterbuch heißt TEXT. Das Programm wird mit TEXT 1 gestartet. Das Programm wurde auf der Grundlage des entsprechenden C-Programms unter Verwendung der elementaren Byteoperationen entwickelt.

8. Zusammenfassung

Ausgehend von der Darstellung der historischen Entwicklung, wurden fünf für die Mikrorechnerprogrammierung wichtige Sprachen etwas ausführlicher betrachtet. Das Beispiel zur Textverarbeitung aus Heft 3 wurde in allen Sprachen implementiert.
Aus den Darlegungen ist zu ersehen, daß es die ideale Programmiersprache noch nicht gibt und in der nächsten Zukunft auch nicht geben wird. Durch die Vielfalt der objektiven und subjektiven Bedingungen hat jede Programmiersprache ihre Existenzberechtigung, und es lohnt sich, sich mit jeder Programmiersprache näher zu beschäftigen. Dieser Beitrag soll dabei dem Amateur auch eine kleine Hilfe geben, »seine« Programmiersprache auszuwählen.

Autor:
Doz. Dr. sc. techn. Thomas Horn

Hochschuldozent
an der Sektion Informationsverarbeitung
der Ingenieurhochschule Dresden

Mikrorechnergesteuerte Analog-Digital-Umsetzung mit dem Polycomputer 880

Einleitung

Mit dem Polycomputer 880 ist ein vollständiger Mikrorechner mit der CPU U 880 mit 2 KByte ROM-Kapazität (U 505) für das Betriebssystem, 1KByte RAM-Kapazität (U 202), mit zwei Peripheriebausteinen PIO U 855 und der CTC U 857 auf dem Markt, der sich als Mikrorechner-Lernsystem für eigene Programmentwicklungen sowie für einfache Prozeßsteuerungsaufgaben sinnvoll einsetzen läßt.
In der Literatur sind erste Erfahrungen [1], [2], [3] über die Möglichkeiten des Einsatzes dieses ersten in der DDR industriell gefertigten Mikrorechner-Lernsystems angegeben. Während in [4] Zusatzgeräte für den Polycomputer 880 vorgestellt sind, wurde im Heft 3 dieser Broschürenreihe [5] dieses Gerät hardwaremäßig eingehend beschrieben. Im folgenden soll die Möglichkeit der Anwendung des Polycomputers 880 für eine Analog-Digital-Umsetzung und deren grafische Darstellung auf einem Sichtgerät für Demonstrationszwecke der Funktionsweise einfacher ADUs unter Verwendung einer einfachen Programmierung der Umsetzerfunktion dargestellt werden. Im einzelnen werden die Funktionen des Stufen- oder Folge-ADUs und des ADUs nach dem Prinzip der sukzessiven Approximation vorgestellt.



Bild 1. Blockschaltbild der Hardwarebaugruppen zur Realisierung der ADU-Funktionen

In Bild 1 ist das prinzipielle Blockschaltbild des Aufbaus für die Realisierung der ADU-Funktionsweisen gegeben.
Die Wirkungsweise dieses Aufbaus ist folgende: Die zu messende und darzustellende analoge Spannung U_M wird über eine Tast- und Halteschaltung auf den Eingang eines Komparators gegeben und mit einem vom Mikrorechner bereitgestellten und über die PIO ausgegebenen 8-bit-Datenwort, das mittels eines DAU zu einer Vergleichsspannung U_V gewandelt wird, verglichen. Im Ergebnis des Spannungsvergleichs steht am Ausgang des Komparators die Information $U_M \geqq U_V$ durch H bzw. $U_M \leqq U_V$ durch L als Eingangsinformation für den Mikrorechner zur Verfügung, die als Auswertekriterium für die softwaremäßige Realisierung der entsprechenden ADU-Algorithmen Verwendung findet. Nach später dargelegten Algorithmen liefert der Mikrorechner ein neues Datenwort,

das über den DAU erneut in eine äquivalente Vergleichsspannung gewandelt und wiederum an den Komparatoreingang gelegt wird. Dieser Zyklus wird softwaremäßig gesteuert und so lange wiederholt, bis die Vergleichsspannung U_V vom DAU innerhalb der kleinsten Digitalisierungsstufe mit der Meßspannung U_M übereinstimmt.
Die einzelnen Schritte bzw. Stufen der Umsetzung können durch geeignete Wahl der Umsetzungsgeschwindigkeit sowohl digital auf der Anzeige des Mikrorechners als auch als grafische Darstellung auf dem Sichtgerät verfolgt werden.

Funktionsweise der AD-Wandlung nach dem Prinzip der Stufenumsetzung bzw. nach dem Verfahren der sukzessiven Approximation



Bild 2. Prinzip des Stufen- oder Folge-ADU

Bei der AD-Wandlung nach dem Stufen- oder Folgeverfahren (Bild 2) wird eine angelegte Spannung mit einer stufenweise (schrittweise) veränderlichen Vergleichsspannung U_V verglichen, wobei die Spannungsstufung durch den Spannungsbereich, der umgesetzt werden soll, und durch die Anzahl der Bits des Datenworts vorgegeben ist. Die Anstiegsgeschwindigkeit bzw. Umsetzgeschwindigkeit ist dabei durch die Größe der kleinsten Spannungsstufe und durch die Taktfrequenz bestimmt.

Nachdem die stufenförmig ansteigende Vergleichsspannung U_V die Meßspannung U_M erreicht hat, wechselt bei konstanter Meßspannung das Signal am Komparatorausgang ständig zwischen H und L, d.h., es wird abwechselnd eine Spannungsstufe zu- bzw. abgeschaltet. Ändert sich zeitlich die Meßspannung, so »folgt« der Folge-ADU unterhalb der hardware- bzw. softwarebedingten Grenzfrequenz der angelegten Spannung. Um zu verhindern, daß bei konstant bleibender Eingangsspannung die Vergleichsspannung ständig um die kleinste Spannungsstufung schwankt, kann einerseits der Komparator mit einer Schalthysterese von $\pm 1/2$ Spannungsstufe ΔU versehen werden bzw. andererseits ein DA-Wandler verwendet werden, der eine um 1 bit höhere Auflösung besitzt, als digitale Ausgänge herausgeführt sind.



Bild 3. Prinzip der sukzessiven Approximation

In Bild 3 ist das Prinzip der AD-Wandlung nach der sukzessiven Approximation verdeutlicht. Der Wandler besteht wieder – wie in Bild 1 gezeigt – aus einem Komparator, dem internen DA-Wandler und einer im Mikrorechner softwaremäßig realisierten Approximationslogik. Die vorgeschaltete Tast- und Halteschaltung ist nur erforderlich, wenn sich U_M während der Umsetzzeit um $\geqq 1/2$ LSB ändert (LSB = least significant bit = kleinste unterscheidbare Amplitudenstufe).
Mit dem Startimpuls wird ein MSB (MSB = most significant bit = Stufe mit der höchsten Wertigkeit = $U_{max/2}$) gesetzt und damit das Vergleichssignal auf $U_{max/2}$ eingestellt. Der Komparator vergleicht U_M und U_V. Da für das MSB, wie in Bild 3 dargestellt, $U_V < U_M$ ist, liegt der Ausgang auf H, und mit der nächsten H/L-Flanke des Taktes wird das MSB »verriegelt«, d.h., es bleibt gesetzt. Mit dem zweiten Taktimpuls wird MSB/2 = $U_{max/4}$ dazugeschaltet, und der Komparator vergleicht jetzt U_M mit U_V = MSB + MSB/2 = $U_{max/2} + U_{max/4}$.
Solange die Vergleichsspannung $U_V < U_M$ ist, steht am Komparatorausgang H an, und die entsprechenden Spannungsstufungen werden »verriegelt« bzw. bleiben gesetzt. Sobald sich bei der bitweisen Aufsummierung der gestuften, immer weiter halbierten Spannungswerte eine Vergleichsspannung $U_V > U_M$ ergibt, steht am Komparatorausgang L an, und das letzte bit wird zurückgesetzt. Dieser Vorgang wiederholt sich, bis das LSB abgearbeitet ist.
Diese Wandlerprinzipien werden durch ein kleines Unterprogramm des Anwenderprogramms des Mikrorechners realisiert, wobei beim Folge-ADU in Abhängigkeit vom Vergleich U_M mit U_V eine kleinste Spannungsstufe ΔU aufsummiert oder subtrahiert werden muß. Beim ADU nach der sukzessiven Approximation erfolgt diese Aufsummation oder Subtraktion mit gestuften (d.h. ausgehend von der maximalen Umsetzspannung U_{max} jeweils halbierten) Spannungswerten.

Schaltungsbeschreibung

Durch den Einsatz des Polycomputer 880 als Steuer- und Verarbeitungssystem kann der Hardwareaufwand relativ minimal gehalten werden. So enthält die Schaltung nach Bild 4, die auf einer Leiterkarte realisiert an die entsprechende Steckerleiste des Polycomputers angeschlossen wird, im wesentlichen nur zwei Baugruppen:

- einen 8-bit-DAU mit Spannungsverstärker
- einen Komparator.

Damit lassen sich u.a. folgende Funktionsweisen realisieren:

- Digital-Analog-Umsetzung
- digital steuerbarer Funktionsgenerator
- Analog/Digital-Umsetzung (Einquadrantenumsetzung) mit verschiedenen Umsetzverfahren
- Nullpunkttrigger bzw. Trigger mit Offsetkompensation
- Fensterdiskriminator
- Zeit- und Frequenzmessung.

Bedingt durch den vordergründigen Einsatz in Demonstrations- und Lernsysteme wurden die einzelnen Schaltungen unkompliziert und übersichtlich gehalten und weisen keine speziellen Besonderheiten auf.
Der Digital-Analog-Umsetzer arbeitet nach dem Prinzip der Stromsummation, wobei die einzelnen Stromquellen durch die Widerstände $R_{V1} \cdots R_{V8}$ realisiert werden und die Stromsenke der Widerstand R_8 bildet. Die Steuerung der Stromquellen wird über Kurzschlußschalter (Variante mit Open-Kollektor-Inverter) bzw. über Serienkurzschlußschalter (Variante mit Gegentakt-Inverter) erreicht. Diese einfache Schaltungsvariante besitzt den Vorteil, daß durch Parameteränderungen der einzelnen Widerstände



Bild 4. 8-bit-DAU mit Spannungsverstärker und Komparator

$R_{V1} \cdots R_{V8}$ bzw. R_S deren Einfluß auf Kennlinien- und Linearitätsfehler dargestellt werden kann. Weiterhin ist die Möglichkeit gegeben, durch Änderung der Wichtung der Widerstandswerte $R_{V1} \cdots R_{V8}$ nichtlineare Umsetzcharakteristiken zu erzeugen.

Dem Digital-Analog-Wandler schließt sich ein Spannungsverstärker (MAA 741) an, mit dem die Größe der Ausgangsspannung U_A variiert werden kann. Zur Unterdrückung von Glitsches wurde der Tiefpaß R_3/C_2 am Ausgang des DAU vorgesehen.

Der zweite Schaltungsteil beinhaltet einen Komparator A 110, wobei ein Eingang mit dem DAU-Ausgang verbunden ist, und der andere über einen Spannungsteiler das Eingangssignal erhält (siehe auch Bild 1).

Die Steuerung des DAU bzw. die Informationsübernahme des Komparatorzustandes erfolgt über die PIO U 855 D, die bereits im Polycomputer 880 enthalten ist. Desgleichen erfolgt die Stromversorgung der Zusatzschaltungen durch den Mikrorechner über den Peripheriesteckverbinder.

Programmbeschreibung

Hauptprogramm

Vor dem Programmstart sind die Register A mit der Zeitkonstanten für die CTC, B mit der Schrittweite und C mit der Anfangsbitbelegung für den DAU zu laden. Während der Anfangsinitialisierung des Programms werden diese Registerinhalte auf Merkzellen für die spätere Bearbeitung gerettet. Falls im Akku eine "Ø" übergeben wurde, wird diese in eine "1" übergeführt, da die CTC die "Ø" als "ØFFH" interpretieren würde, also die maximal mögliche Zeitkonstante. Das I-Register

und damit der höherwertige Teil des Interruptvektors wird mit 42 H geladen sowie der Interruptmode 2 eingestellt.

Für die Datenbyteausgabe an den DAU wird, wie im Stromlaufplan ersichtlich, Port A der PIO genutzt und deshalb für Mode Ø »Byte-Ausgabe« initialisiert. Das Komparatorbit ist an Bit Ø des Port B der PIO angeschlossen. Port B wird hier für Mode 3 »Bitgesteuerte E/A« initialisiert, wobei alle Leitungen als Eingänge definiert sind.

Der PIO-Initialisierung folgt die Ausgabe des Programmnamens »ADU« auf dem LED-Display. Bis hierher kann man die Befehlsfolge als Initialisierungsteil des Programms bezeichnen. Es schließt sich der eigentliche Meßzyklus an. Dazu wird die aktuelle Bitbelegung des Ausgabebytes von der Zelle »Merk1« geladen und über die PIO an den DAU ausgegeben sowie die CTC für die Realisierung des Zeitrasters initialisiert. Der Wert des Ausgabebytes ⟨C⟩ wird zur Kompensation der Wirkung des Eingangsspannungsteilers mit 4 multipliziert, in eine Dezimalzahl gewandelt und zur Anzeige gebracht.

Die CPU begibt sich nun in eine Schleife und wartet den CTC-Interrupt ab. Innerhalb der Interruptwarteschleife wird die LED-Anzeige durch zyklische Ausgabe ständig aufgefrischt. Die Betriebsart des ADUs wird innerhalb der CTC-Interrupt-Serviceroutine mit der Befehlsfolge des »USER-Programms« festgelegt (siehe Bild 5).

CTC-Interrupt-Serviceroutine

Empfängt die CPU einen CTC-Interrupt, so wird der Programmzeiger der CPU auf die Interrupt-Serviceroutine gestellt und diese abgearbeitet. Die CTC wird abgeschaltet und das Vergleichsergebnis des Komparators in



Bild 5. Hauptprogramm

den Akku eingelesen. Anschließend werden das aktuelle Ausgabebitmuster und die Schrittweite von der Zelle »Merk1« in das BC-Register geladen. Damit ist das »USER-Programm« mit den notwendigen Daten versorgt. Für die sukzessive Approximation tritt als Grenzfall der Wert $B = 1$ auf. Damit auch bei dieser Betriebsart nach Erreichen des Komparatorpunkts weitere Messungen ausgelöst werden, bewirkt das Programm nach Abarbeiten einer Hilfszeitschleife das Rücksetzen

der Arbeitszellen und einen Neubeginn der Approximation.
Nachdem im eingelesenen Vergleichsergebnis des Komparators die überflüssigen Bits ausgeblendet wurden, kann mit dem »USER-Programm« die gewünschte Arbeitsweise des ADU realisiert werden. Vor dem anschließenden Rücksprung aus der Interruptroutine wird die Zieladresse unter Zuhilfenahme von Austauschbefehlen in den Stack geladen (Bild 6).



Bild 6. CTC-Interrupt-Serviceroutine

PIO-Initialisierung

Mit diesem Unterprogramm (Bild 7) wird die am E/A-Stecker angeschlossene PIO des Polycomputers so initialisiert, daß über Port A die Ausgabe des Bitmusters an den DAU erfolgen und über Port B das Vergleichsergebnis des Komparators in den Rechner eingelesen werden kann. Da ein Interruptbetrieb der PIO unnötig ist, wird dieser verboten.



Bild 7. Unterprogramm 1
PIO-Initialisierung

CTC-Initialisierung

Zur Erzeugung des Zeitrasters, unter welchem der DAU arbeiten soll, wird der Kanal 2 als Zeitgeber und der Kanal 3 als Zähler initialisiert. Diese Kombination ist möglich, weil im Polycomputer der Zählerausgang des Kanals 2 mit dem Triggereingang des Kanals 3 verbunden ist. Damit kein vorzeitiger Interrupt ausgelöst wird, wird die CPU vor der Initialisierung in den Dissable-Interrupt-Zustand gebracht.
Bei der Wahl des Interruptvektors ist zu beachten, daß die Interrupttabelle der CTC auf einer durch acht teilbaren Adresse beginnen muß und dieser nur über Kanal Ø der CTC übermittelt werden kann (Bild 8).



Bild 8.
Unterprogramm 4
CTC-Initialisierung



Bild 9.
Unterprogramm 2
Anzeige der Überschrift

Anzeige der Überschrift

Um den Programmnamen »ADU« auf der LED-Anzeige des Polycomputers ausgeben zu können, ist es zweckmäßig, den Anzeigepuffer des Monitors zu benutzen, da auf diese Weise gleich das interne Treiberprogramm für die Ausgabe genutzt werden kann. Dazu wird die Adresse des Anzeigepuffers in das HL-Register geladen und unter Weiterstellen dieses Zeigers der Puffer mit den entsprechenden Zeichen gefüllt (Bild 9). Zu beachten ist, daß im Polycomputer 880 keine ASCII-Zeichen benutzt werden.



Bild 10.
Unterprogramm 3
Ausgabe DAU

Ausgabe DAU

Die aktuelle Bitbelegung des Ausgabebytes an den DAU steht an dieser Stelle des Programms im C-Register. Als Treiber zur Ansteuerung des Widerstandsnetzwerkes des DAU wird der Schaltkreis D1ØØ benutzt. Da dieser die Signale negiert, für die Anschaulichkeit der Arbeitsweise ein highaktives Signal aber am PIO-Ausgang günstiger ist, wird der Inhalt des C-Registers vor der Ausgabe negiert (Bild 10).

Multiplikation × 4 und Konvertierung Hexadezimal — Dezimal

Nachdem das B-Register zum Auffangen des Überlaufs der sich anschließenden Linksrotation gelöscht ist, werden die Bits Ø, 1 und 2 aus dem C-Register ausgeblendet. Diese Maßnahme ist notwendig, da erstens die zwei niederwertigsten Bits im DAU nicht verdrahtet sind und zweitens ohne die Ausblendung von Bit 2 die LED-Anzeige ständig zwischen zwei Werten wechseln würde. Die zweimalige Linksrotation des C-Registers bewirkt eine Multiplikation mit vier. Damit wird im Zusammenhang mit der verwendeten Hardwarelösung der Meßbereich Ø ··· 10 V eingestellt.
Da der so erhaltene Meßwert jedoch auf Grund der Arbeitsweise des Mikrorechners und des DAU als Hexadezimalzahl im BC-Register steht, der Mensch aber gewohnt ist, den Meßwert im Dezimalsystem zu erfassen, übernimmt eine Konvertierungsroutine die Wandlung in eine Dezimalzahl (Bild 11). Der Algorithmus der Konvertierung beruht auf der zyklischen Subtraktion von Zehnerpotenzen, beginnend mit der höchsten Wertigkeit, wobei die mitgezählten Zyklusanzahlen die gesuchte Dezimalzahl darstellen.



Bild 11. Unterprogramm 5
Multiplikation × 4 und Konvertierung Hexadezimal-Dezimal

Anzeigepuffer laden

Da der Polycomputer nicht wie normalerweise üblich mit ASCII-Zeichen arbeitet, wurde zur Wandlung des Meßwerts in den Anzeigecode der 7segmentanzeige die interne Referenztabelle benutzt. Der höherwertige Teil der Tabellenadresse liegt mit "3" fest. Der niederwertige Teil der Adresse ergibt sich aus dem jeweiligen Zahlenwert plus einem Festwert von 1Ø H. Der unter dieser Adresse aufgefundene Anzeigecode wird fortlaufend in den Anzeigepuffer eingeschrieben. Bei der ersten und dritten Stelle muß vor dem Abspeichern noch der Dezimalpunkt ausgeblendet werden (Bild 12).



Bild 12. Unterprogramm 6
Anzeigepuffer laden

Anzeige der Daten auf dem LED-Display

Dieses Programm existiert bereits im Polycomputer als Unterprogramm des Monitors und wird vom ADU-Programm mitgenutzt. Eine Programmbeschreibung ist in der Dokumentation zum Polycomputer zu finden [6].
Das detaillierte Anwenderprogramm zur Realisierung der ADU-Funktionen mit Hilfe des Polycomputer 880 finden Sie auf S. 44 bis 48.

Inbetriebnahme der mikrorechnergesteuerten ADU

Nachdem das auf S. 44 dargestellte Anwenderprogramm mittels Tastatur eingegeben bzw., falls es bereits auf einer Tonbandkassette abgespeichert ist, in den RAM-Bereich des Mikrorechners eingelesen ist (Anfangsadresse 4000, Endadresse 4300), kann das folgende Unterprogramm für den Stufen- oder Folge-ADU über die Tastatur im Hexadezimalcode ab Adresse 4002 eingegeben werden:

Mnemonische Beschreibung		Adresse	Maschinenprogramm	
CMP	A	4ØØ2	FE	ØØ
JPZ	M1	4ØØ4	CA	ØB 4Ø
LD	A, C	4ØØ7	79	
SUB	B	4ØØ8	9Ø	
LD	C, A	4ØØ9	4F	
RET		4ØØA	C9	
M1: LD	A, C	4ØØB	79	
ADD	B	4ØØC	8Ø	
LD	C, A	4ØØD	4F	
RET		4ØØE	C9	

In diesem Unterprogramm wird jeweils in einer Additions- bzw. Subtraktionsschleife der aktuelle Inhalt des C-Registers in den Akkumulator A geladen und mit dem im B-Register abgespeicherten Wert der Spannungsstufung (im Fall des Folge-ADU wird die kleinste Spannungsstufe $\Delta U = Ø4H$ gewählt, da im realisierten DAU die beiden niederwertigen Bit ØlH und Ø2H nicht belegt sind) erhöht bzw. erniedrigt. Diese stufenweise Annäherung der »Vergleichsspannung« an die Meßspannung U_M kann auch einfach dadurch realisiert werden, daß man anstelle der Additionsschleife den Befehl INC C = ØCH bzw. für die Subtraktionsschleife DEC C = ØDH einfügt. Dabei erhöht dann der Rechner die »Vergleichsspannung« in Ø1H-Stufen, während extern auf dem Sichtgerät die Spannungsstufung in Ø4H-Schritten angezeigt wird.
Zur Realisierung der Funktion des ADU nach dem Verfahren der sukzessiven Approximation, die gleichfalls durch eine Additions- und Subtraktionsschleife verwirklicht werden kann, ist nur zu gewährleisten, daß der im Register B abgespeicherte Anfangswert $U_{max/2}$ vor jeder Schleifenabarbeitung halbiert wird. Dies ist einfach durch den Befehl RRC B = CB Ø8, der auf die Adresse 4ØØØ und 4ØØ1 abgespeichert wird, zu erreichen, da durch die Rechtsrotation eines Bits die mathematische Division durch 2 der Spannungsstufung im Register B realisiert wird. Durch ein einfaches Ersetzen dieses RRC-Befehls durch NOP-Befehle wird die Funktion des Stufen- oder Folge-ADU's wieder hergestellt.
Während vor dem Start des Programms und damit der Spannungsumsetzung im Register C ein beliebiger Anfangswert (z.B. Ø4H für den Folge-ADU bzw. 8ØH für die sukzessive Approximation) eingegeben wird, ist durch die Eingabe einer Hexadezimalzahl in das A-Register die Umsetzgeschwindigkeit wählbar. Je größer diese Zahl gewählt wird, desto größer ist die Zeitschleife und desto kleiner die Umsetzgeschwindigkeit. Bei Eingabe von

```
                        ABS. ASSEMBLER  MEOS 1521 V3.2/4
  LOC        OBJ.CODE     STMT     SOURCE PROGRAM
=HEX==DEC==========================================================================18/09/84==001=============
                            00001          PN   ADU
                            00002  ;
                            00003  ;
                            00004  ;       ANALOG - DIGITALWANDLUNG MITTELS POPLYCOMP - 880
                            00005  ;       UMSETZZEITGRENZE IST SOFTWAREBEDINGT
                            00006  ;
                            00007  ;       10.09.84
                            00008  ;
                            00009          ORG  4100H
                            00010  ;
                            00011  ;
4100 16640  FE 00           00012  START: CMP  0
4102 16642  20 02           00013          JRNZ ST1-#
4104 16644  3E 01           00014          LD   A,1
4106 16646  32 62 42        00015  ST1:    LD   (MERK0),A          ;ZEITKONSTANTE FUER CTC
4109 16649  3E 0A           00016          LD   A,0AH
410B 16651  32 65 42        00017          LD   (MERK2),A          ;HILFSZEITKONSTANTE
410E 16654  3E 42           00018          LD   A,42H
4110 16656  ED 47           00019          LD   I,A                ;HWT - INTERRUPTVEKTOR
4112 16658  ED 5E           00020          IM2                     ;INTERRUPT - MODE 2
4114 16660  ED 43 63 42     00021          LD   (MERK1),BC         ;ANFANGSWERT + SCHRITTWEITE FUER UMSETZUNG
4118 16664  CD 6C 41        00022          CALL PIOI               ;PIO - INITIALISIEREN
411B 16667  CD 98 41        00023          CALL ANZLO              ;ANZEIGE LOESCHEN
411E 16670  ED 4B 63 42     00024  M0:     LD   BC,(MERK1)
4122 16674  CD A7 41        00025          CALL PIOAU              ;AUSGABE AN DAU
4125 16677  CD 81 41        00026          CALL CTCI               ;CTC - INIT.
4128 16680  CD E8 41        00027          CALL KONV               ;MULTIPLIKATION * 4 UND KONVERTIERUNG HEX --> DEZ
412B 16683  CD AD 41        00028          CALL ANZLA              ;ANZEIGEPUFFER LADEN
412E 16686  11 58 42        00029  M1:     LD   DE,ANZBE           ;(DE) = ANZEIGEPUFFER
4131 16689  F3              00030          DI                      ;INTERRUPT FUER CPU VERBIETEN
4132 16690  CD 4E 01        00031          CALL ANZEI              ;ANZEIGEN DES AKTUELLEN WERTES.
4135 16693  FB              00032          EI                      ;INTERRUPT FUER CPU ERLAUBEN
4136 16694  18 F6           00033          JR   M1-#
                            00034  ;
                            00035  ;       CTC - INTERRUPT - SERVICEROUTINE
                            00036  ;
4138 16696  3E 27           00037  INT:    LD   A,27H
413A 16698  D3 89           00038          OUT  89H                ;CTC - AUSSCHALTEN
413C 16700  DB 86           00039          IN   86H                ;KOMPARATOR (BIT 0)
413E 16702  ED 4B 63 42     00040          LD   BC,(MERK1)
4142 16706  57              00041          LD   D,A                ;AKKU RETTEN
4143 16707  78              00042          LD   A,B
4144 16708  FE 01           00043          CMP  1
```

```
4146 16710  20 13           00044          JRNZ INT1-#
4148 16712  3A 65 42        00045          LD   A,(MERK2)
414B 16715  3D              00046          DEC  A                  ;HILFSZEITSCHLEIFE
414C 16716  32 65 42        00047          LD   (MERK2),A
414F 16719  20 10           00048          JRNZ INT2-#
4151 16721  01 80 80        00049          LD   BC,8080H
4154 16724  3E 0A           00050          LD   A,0AH
4156 16726  32 65 42        00051          LD   (MERK2),A
4159 16729  18 06           00052          JR   INT2-#
415B 16731  7A              00053  INT1:   LD   A,D
415C 16732  E6 01           00054          AND  1                  ;KOMPARATORBIT AUSBLENDEN
415E 16734  CD 00 40        00055          CALL USER
4161 16737  ED 43 63 42     00056  INT2:   LD   (MERK1),BC         ;NEUES AUSGABE - BYTE AN DAU
4165 16741  E3              00057          EX   (SP),HL
4166 16742  21 1E 41        00058          LD   HL,M0              ;RUECKSPRUNGADRESSE IN STACK LADEN
4169 16745  E3              00059          EX   (SP),HL
416A 16746  ED 4D           00060          RETI
                            00061          EJEC
                            00062  ;
                            00063  ;       PIO - INITIALISIERUNG
                            00064  ;
416C 16748  3E 03           00065  PIOI:   LD   A,3                ;INT. SPERRE,KANAL A
416E 16750  D3 85           00066          OUT  85H
4170 16752  3E 0F           00067          LD   A,0FH              ;BYTE - AUSGABE
4172 16754  D3 85           00068          OUT  85H
4174 16756  3E 03           00069          LD   A,3                ;INT. SPERRE,KANAL B
4176 16758  D3 87           00070          OUT  87H
4178 16760  3E CF           00071          LD   A,0CFH             ;MODE - 3
417A 16762  D3 87           00072          OUT  87H
417C 16764  3E FF           00073          LD   A,0FFH             ;ALLE LEITUNGEN SIND EINGAENGE
417E 16766  D3 87           00074          OUT  87H
4180 16768  C9              00075          RET
                            00076  ;
                            00077  ;       CTC - INITIALISIERUNG
                            00078  ;
4181 16769  F3              00079  CTCI:   DI
4182 16770  3E 07           00080          LD   A,7                ;STEUERWORT
4184 16772  D3 8A           00081          OUT  8AH
4186 16774  3E FF           00082          LD   A,0FFH             ;ZEITKONSTANTE FUER KANAL 2
4188 16776  D3 8A           00083          OUT  8AH
418A 16778  3E 50           00084          LD   A,50H              ;HWT - INTERRUPTVEKTOR FUER CTC
418C 16780  D3 88           00085          OUT  88H                ;(KANN NUR UEBER KANAL - 0 AUSGEGEBEN WERDEN)
418E 16782  3E F7           00086          LD   A,0F7H             ;STEUERWORT
4190 16784  D3 8B           00087          OUT  8BH
4192 16786  3A 62 42        00088          LD   A,(MERK0)          ;ZEITKONSTANTE
4195 16789  D3 8B           00089          OUT  8BH
4197 16791  C9              00090          RET
```

```
                                00091  ;
                                00092  ;       ANZEIGE LOESCHEN
                                00093  ;
4198 16792  21 58 42            00094  ANZLO: LD   HL,ANZBE            ;ANZEIGEPUFFER
419B 16795  36 F5               00095         LD   M,0F5H             ;A
419D 16797  23                  00096         INC  HL
419E 16798  36 97               00097         LD   M,97H              ;D
41A0 16800  23                  00098         INC  HL
41A1 16801  36 C7               00099         LD   M,0C7H             ;U
41A3 16803  23                  00100         INC  HL
41A4 16804  36 00               00101         LD   M,0                ;SPACE
41A6 16806  C9                  00102         RET
                                00103  ;
                                00104  ;       AUSGABEBYTE AN ADU
                                00105  ;
41A7 16807  79                  00106  PIOAU: LD   A,C
41A8 16808  ED 44               00107         NEG
41AA 16810  D3 84               00108         OUT  84H
41AC 16812  C9                  00109         RET
                                00110         EJEC
                                00111  ;
                                00112  ;       LADEN DES ANZEIGEPUFFERS
                                00113  ;
41AD 16813  21 5C 42            00114  ANZLA: LD   HL,ANZBE+4
41B0 16816  16 03               00115         LD   D,3                ;HWT - TABELLENADRESSE FUER  7-SEGMENTANZEIGE
41B2 16818  78                  00116         LD   A,B
41B3 16819  E6 F0               00117         AND  0F0H
41B5 16821  FE 00               00118         CMP  0
41B7 16823  28 08               00119         JRZ  AN1-#
41B9 16825  0F                  00120         RRCA
41BA 16826  0F                  00121         RRCA
41BB 16827  0F                  00122         RRCA
41BC 16828  0F                  00123         RRCA
41BD 16829  C6 10               00124         ADD  10H                ;TABELLENADRESSE BILDEN
41BF 16831  5F                  00125         LD   E,A                ;HWT - TABELLENADRESSE
41C0 16832  1A                  00126         LD   A,(DE)
41C1 16833  E6 F7               00127  AN1:   AND  0F7H              ;DEZIMALPUNKT AUSBLENDEN
41C3 16835  77                  00128         LD   M,A
41C4 16836  23                  00129         INC  HL
41C5 16837  78                  00130         LD   A,B
41C6 16838  E6 0F               00131         AND  0FH
41C8 16840  C6 10               00132         ADD  10H                ;TABELLENADRESSE BILDEN
41CA 16842  5F                  00133         LD   E,A                ;HWT - TABELLENADRESSE
41CB 16843  1A                  00134         LD   A,(DE)
41CC 16844  77                  00135         LD   M,A
41CD 16845  23                  00136         INC  HL
```

```
41CE 16846  79                  00137         LD   A,C
41CF 16847  E6 F0               00138         AND  0F0H
41D1 16849  0F                  00139         RRCA
41D2 16850  0F                  00140         RRCA
41D3 16851  0F                  00141         RRCA
41D4 16852  0F                  00142         RRCA
41D5 16853  C6 10               00143         ADD  10H                ;TABELLENADRESSE BILDEN
41D7 16855  5F                  00144         LD   E,A                ;HWT - TABELLENADRESSE
41D8 16856  1A                  00145         LD   A,(DE)
41D9 16857  E6 F7               00146         AND  0F7H               ;DEZIMALPUNKT AUSBLENDEN
41DB 16859  77                  00147         LD   M,A
41DC 16860  23                  00148         INC  HL
41DD 16861  79                  00149         LD   A,C
41DE 16862  E6 0F               00150         AND  0FH
41E0 16864  C6 10               00151         ADD  10H                ;TABELLENADRESSE BILDEN
41E2 16866  5F                  00152         LD   E,A                ;HWT - TABELLENADRESSE
41E3 16867  1A                  00153         LD   A,(DE)
41E4 16868  E6 F7               00154         AND  0F7H               ;DEZIMALPUNKT AUSBLENDEN
41E6 16870  77                  00155         LD   M,A
41E7 16871  C9                  00156         RET
                                00157  ;
                                00158  ;       MULTIPLIKATION * 4 DES C - REGISTERS
                                00159  ;
41E8 16872  06 00               00160  KONV:  LD   B,0
41EA 16874  79                  00161         LD   A,C
41EB 16875  E6 F8               00162         AND  0F8H               ;BITS 0,1,2 AUSBLENDEN
41ED 16877  4F                  00163         LD   C,A
41EE 16878  CB 21               00164         SLA  C
41F0 16880  CB 10               00165         RL   B
41F2 16882  CB 21               00166         SLA  C
41F4 16884  CB 10               00167         RL   B
                                00168         EJEC
                                00169  ;
                                00170  ;       WANDLUNG HEX --> DEZ
                                00171  ;
41F6 16886  60                  00172         LD   H,B
41F7 16887  69                  00173         LD   L,C
41F8 16888  01 00 00            00174         LD   BC,0
41FB 16891  11 E8 03            00175         LD   DE,1000
41FE 16894  AF                  00176         XOR  A
41FF 16895  04                  00177  H1:    INC  B
4200 16896  ED 52               00178         SBC  HL,DE
4202 16898  30 FB               00179         JRNC H1-#
4204 16900  19                  00180         ADD  HL,DE
4205 16901  05                  00181         DEC  B
4206 16902  CB 20               00182         SLA  B
```

```
4208 16904 CB 20          00183          SLA  B
420A 16906 CB 20          00184          SLA  B
420C 16908 CB 20          00185          SLA  B
420E 16910 11 64 00       00186          LD   DE,100
4211 16913 AF             00187          XOR  A
4212 16914 04             00188    H0:   INC  B
4213 16915 ED 52          00189          SBC  HL,DE
4215 16917 30 FB          00190          JRNC H0-#
4217 16919 19             00191          ADD  HL,DE
4218 16920 05             00192          DEC  B
4219 16921 11 0A 00       00193          LD   DE,10
421C 16924 AF             00194          XOR  A
421D 16925 0C             00195    H3:   INC  C
421E 16926 ED 52          00196          SBC  HL,DE
4220 16928 30 FB          00197          JRNC H3-#
4222 16930 19             00198          ADD  HL,DE
4223 16931 0D             00199          DEC  C
4224 16932 CB 21          00200          SLA  C
4226 16934 CB 21          00201          SLA  C
4228 16936 CB 21          00202          SLA  C
422A 16938 CB 21          00203          SLA  C
422C 16940 79             00204          LD   A,C
422D 16941 B5             00205          OR   L
422E 16942 4F             00206          LD   C,A
422F 16943 C9             00207          RET
                          00208    ;
                          00209          ORG  4250H
                          00210    ;
                          00211    ;     INTERRUPTTABELLE FUER CTC - BAUSTEIN
                          00212    ;
4250 16976 38 41          00213    INTA: DW   INT                ;KANAL 0
4252 16978 38 41          00214          DW   INT                ;KANAL 1
4254 16980 38 41          00215          DW   INT                ;KANAL 2
4256 16982 38 41          00216          DW   INT                ;KANAL 3
                          00217    ;
4258 16984                00218    ANZBE: DEF  10                ;ANZEIGEPUFFER
4262 16994                00219    MERK0: DEF  1                 ;ZEITKONSTANTE - CTC
4263 16995                00220    MERK1: DEF  2                 ;AUSGABEBYTE AM DAU
4265 16997                00221    MERK2: DEF  1                 ;HILFSZEITKONSTANTE
4266 16998                00222          DEF  0
                          00223    ;
                          00224    ANZEI: EQU  14EH
                          00225    USER:  EQU  4000H
                          00226          END
                          PROGRAM CONTAINS 000000 ERRORS
```


ØØH sind allein die im Anwenderprogramm programmierten Zeitschleifen wirksam.

Abschließend sei festgestellt, daß das Auflösungsvermögen dieser mikrorechnergestützten Analog-Digital-Wandler durch die Bit-Zahl der Datenworte vorgegeben ist und die Genauigkeit durch Verwendung der nicht speziell stabilisierten Rechnerbetriebsspannung zur Realisierung der Vergleichsspannung bestimmt wird. In [4] sind industrielle Zusatzbaugruppen zur Realisierung einzelner ADU mit hoher Genauigkeit vorgestellt. Ziel dieses Beitrages war es, zu zeigen, wie unter Verwendung des Polycomputers 880 mit sehr geringem Hardwareaufwand und mit einfachen Softwaremitteln die Funktionsweisen einzelner ADU realisiert und demonstriert werden können.

Literatur

[1] Arnold, H.; Pilz, W.: Polycomputer 880. – In: r f e. – Berlin **31** (1982) 6, – S. 385—386

[2] Jakubaschk, H.: Erfahrungen mit dem Polycomputer PC 880. – In: r f e. – Berlin **32** (1983) 8, – S. 492—493

[3] Burkhardt, S.; Hübner, U.: Technik und Anwendung des Mikrorechnerlernsystems Polycomputer 880. – In: r f e. – Berlin **33** (1984) 5, – S. 282—287

[4] Hübner, U.: Zusatzgeräte für Polycomputer 880. – In: r f e. – Berlin **33** (1984) 7, – S. 415—419

[5] Hübner, U.: Polycomputer 880 – Anwendung und Erweiterungsmöglichkeiten. – In: Kleinstrechner-TIPS, H. 3. – Leipzig, 1985

[6] Dokumentation des Polycomputer 880 (5 Hefte)

Autoren:

Dr.-Ing. Kurt Lehmann
Wissenschaftlicher Oberassistent an der Ingenieurhochschule Dresden

Dipl.-Ing. Lothar Schumann
Wissenschaftlicher Assistent an der Ingenieurhochschule Dresden

Dipl.-Ing. Peter Walke
Problemanalytiker
Bezirkshygieneinspektion und -institut Dresden

Rechentechnische Begriffe für den Laien erklärt

Kode Eineindeutige Abbildung einer Menge von Informationen auf eine Menge von Zahlen. Ein weit verbreitetes Kodesystem ist der ASCII-Kode.

SP Leerzeichen (SPACE). Als Leerzeichen wird ein Leerschritt bei der Ein- und Ausgabe von Informationen (Texten) bezeichnet.

CR Wagenrücklauf (CARRIAGE RETURN). Das Steuerzeichen 'Wagenrücklauf' bewirkt bei Druckern die Rückführung des Druckkopfes an den linken Blattrand und bei Bildschirmgeräten die Rückführung des Cursors an den linken Bildschirmrand.

LF Zeilenvorschub (LINE FEED). Das Steuerzeichen 'Zeilenvorschub' bewirkt auf Druckern und Bildschirmgeräten den Übergang auf die nächste Zeile. Der nächste Zeilenanfang wird eingestellt durch Ausgabe der Steuerzeichen CR und LF, bei manchen Geräten ist dafür auch das Steuerzeichen NL (Neue Zeile – NEW LINE) realisiert.